

**Hinweis:** In den aktuellen Excel-Versionen steht Ihnen der Befehl EINDEUTIG zur Verfügung. In Verbindung mit ANZAHL oder ANZAHL2 können Sie hier einfach die Anzahl unterschiedlicher Einträge in einer Liste ermitteln.

Die folgende Beschreibung hilft Ihnen, wenn der Befehl EINDEUTIG nicht zur Verfügung steht.

## WIE ERMITTELT MAN DIE ANZAHL UNTERSCHIEDLICHER EINTRÄGE IN EINER LISTE?

Wieviel unterschiedliche Einträge enthält die folgende Liste?

	A	B	C
1	<b>lfd. Nr.</b>	<b>Name</b>	
2	001.	Maier	
3	002.	Schmidt	
4	003.	Müller	
5	004.	Walter	
6	005.	Müller	
7	006.	Meier	
8			

Antwort: 5 verschiedene Namen (Meier und Maier sind verschiedene Einträge, Müller gibt es doppelt und zählt als ein Eintrag)

1. Man benötigt eine Matrixfunktion.
2. Mit Hilfe der Funktionen WENN, SUMME, ZÄHLENWENN baut man sich die entsprechende Lösung zusammen.
3. Als Hilfsmittel benötigt man noch den Kehrwert der ZÄHLENWENN-Funktion.
4. Alternativ kann man auch Lösungen mit SUMMENPRODUKT anwenden.
5. Da Listen in der Praxis oft verändert werden, kann man die Lösung auch dynamisieren. Dazu nutzt man am Besten Intelligente Tabellen und strukturierte Verweise.

## LÖSUNG OHNE LEERZELLEN

	A	B	C	D
1	<b>lfd. Nr.</b>	<b>Name</b>		<b>Anzahl verschiedener Elemente</b>
2	001.	Maier		5
3	002.	Schmidt		
4	003.	Müller		
5	004.	Walter		
6	005.	Müller		
7	006.	Meier		
8				
9				
10				
11				

{=SUMME(1/ZÄHLENWENN(B2:B7;B2:B7))}

Die ZÄHLENWENN-Funktion zählt für jede Zeile, wie oft der Name in der Liste vorkommt:

	A	B	C	D
1	<b>lfd. Nr.</b>	<b>Name</b>	<b>ZÄHLENWENN</b>	<b>1/ZÄHLENWENN</b>
2	001.	Maier	1	1
3	002.	Schmidt	1	1
4	003.	Müller	2	0,5
5	004.	Walter	1	1
6	005.	Müller	2	0,5
7	006.	Meier	1	1
8		Summe	8	5
9				

Summiert man die Ergebnisse auf, erhält man ein unsinniges Ergebnis, im Beispiel 8.

Wenn Müller zweimal vorkommt, steht hinter jedem Müller eine 2, bei dreimaligem Vorkommen eine 3 usw., wodurch die Summe für unsere Aufgabe keinen Sinn ergibt. (Spalte C)

Wenn man die Kehrwerte der Ergebnisse in den Zeilen zusammenrechnet, erhält man das richtige Ergebnis. Zwei Müller wären  $\frac{1}{2} + \frac{1}{2} = 1$ , drei Müller  $\frac{1}{3} + \frac{1}{3} + \frac{1}{3} = 1$  usw. (Spalte D). In das Gesamtergebnis geht Müller, egal wie oft er in der Liste vorkommt, als 1 ein.

Wenn man in einer Formel ohne Hilfszellen oder Zwischenberechnung die Anzahl der verschiedenen Einträge berechnet, wird jeder Eintrag in der Liste mit jedem Eintrag verglichen, seine Häufigkeit ermittelt und aus den Kehrwerten die entsprechende Summe gebildet.

**Problem:** Mit einer „normalen“ Funktion funktioniert das nicht. Da wir hier zwei Matrizen miteinander vergleichen - wir vergleichen die Liste der Namen mit der Liste der Namen – benötigen wir eine Matrixfunktion. Diese erhält man, indem man die gesamte Funktion in geschweifte Klammern setzt.

**Wichtig:** Die Klammern darf man nicht mit der Tastatur eintippen. Man schließt die Eingabe der Formel mit der Tastenkombination STRG + SHIFT + ENTER ab. Dadurch entstehen die Klammern.

## LÖSUNG MIT LEERZELLEN

Die bisherige Lösung funktioniert nicht in Listen mit Leerzellen. Hier wird ein mathematischer Fehler erzeugt (Division durch Null).

Jetzt kommt die WENN-Funktion ins Spiel:

Die WENN-Funktion sorgt dafür, dass nur die Zeilen in die Berechnung einbezogen werden, die nicht leer sind.

	A	B	C	D	E
1	<b>lfd. Nr.</b>	<b>Name</b>		<b>Anzahl verschiedener Elemente</b>	
2	001.	Maier		#DIV/0!	
3	002.				
4	003.	Müller			
5	004.	Walter			
6	005.	Müller			
7	006.	Meier			
8					
9				{=SUMME(1/ZÄHLENWENN(B2:B7;B2:B7))}	
10					
11					
12	<b>lfd. Nr.</b>	<b>Name</b>		<b>Anzahl verschiedener Elemente</b>	
13	001.	Maier		4	
14	002.				
15	003.	Müller			
16	004.	Walter			
17	005.	Müller			
18	006.	Meier			
19					
20				{=SUMME(WENN(B13:B18<>"";1/ZÄHLENWENN(B13:B18;B13:B18)))}	
21					

## DYNAMISIERUNG

Wenn sich im bisherigen Beispiel die Liste mit den Namen ändert, d.h. sie wird länger oder kürzer, passt sich unsere Funktion nicht automatisch an. Bei jeder Veränderung der Liste muss man die Funktion manuell anpassen. Das ist äußerst unpraktisch.

Dieses Problem umgehen wir in diesem Beispiel, indem wir aus der vorliegenden Liste eine intelligente Tabelle machen und in der Funktion die Bezüge durch strukturierte Verweise ersetzen.

	A	B	C	D	E	F
1	<b>lfd. Nr.</b>	<b>Name</b>		<b>Anzahl verschiedener Elemente</b>		
2	001.	Maier		5		
3	002.	Schmidt				
4	003.	Müller				
5	004.	Walter				
6	005.	Müller				
7	006.	Meier				
8	007.					
9						
10				{=SUMME(WENN(Tabelle1[Name]<>"";1/ZÄHLENWENN(Tabelle1[Name];Tabelle1[Name])))}		
11						
12	=ZEILE()-ZEILE(lfd_nr)					
13						

Spalte A enthält eine Formel, welche für das automatische Ausfüllen der laufenden Nummer sorgt. Das Zahlenformat ist benutzerdefiniert: 000“.

Wenn man die Markierung in die letzte Zelle der Liste setzt und die TAB-Taste drückt, verlängert sich die Tabelle automatisch, an der Funktion in D2 muss man nichts ändern.